

## G11CAF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

G11CAF returns parameter estimates for the conditional logistic analysis of stratified data, for example, data from case-control studies and survival analyses.

### 2 Specification

```

SUBROUTINE G11CAF(N, M, NS, Z, LDZ, ISZ, IP, IC, ISI, DEV, B, SE,
1          SC, COV, NCA, NCT, TOL, MAXIT, IPRINT, WK, LWK,
2          IFAIL)
  real
1          Z(LDZ,M), DEV, B(IP), SE(IP), SC(IP),
          COV(IP*(IP+1)/2), TOL, WK(LWK)
1          INTEGER
          N, M, NS, LDZ, ISZ(M), IP, IC(N), ISI(N),
1          NCA(NS), NCT(NS), MAXIT, IPRINT, LWK, IFAIL

```

### 3 Description

In the analysis of binary data, the logistic model is commonly used. This relates the probability of one of the outcomes, say  $y = 1$ , to  $p$  explanatory variates or covariates by

$$Prob(y = 1) = \frac{\exp(\alpha + z^T \beta)}{1 + \exp(\alpha + z^T \beta)}$$

where  $\beta$  is a vector of unknown coefficients for the covariates  $z$  and  $\alpha$  is a constant term. If the observations come from different strata or groups,  $\alpha$  would vary from strata to strata. If the observed outcomes are independent then the  $y$ 's follow a Bernoulli distribution, i.e., a binomial distribution with sample size one and the model can be fitted as a generalized linear model with binomial errors.

In some situations the number of observations for which  $y = 1$  may not be independent. For example, in epidemiological research, case-control studies are widely used in which one or more observed cases are matched with one or more controls. The matching is based on fixed characteristics such as age, sex etc. and is designed to eliminate the effect of such characteristics in order to more accurately determine the effect of other variables. Each case-control group can be considered as a stratum. In this type of study the binomial model is not appropriate, except if the strata are large, and a conditional logistic model is used. This considers the probability of the cases having the observed vectors of covariates given the set of vectors of covariates in the strata. In the situation of one case per stratum, the conditional likelihood for  $n_s$  strata can be written as

$$L = \prod_{i=1}^{n_s} \frac{\exp(z_i^T \beta)}{[\sum_{l \in S_i} \exp(z_l^T \beta)]} \quad (1)$$

where  $S_i$  is the set of observations in the  $i$ th stratum, with associated vectors of covariates  $z_l$ ,  $l \in S_i$ , and  $z_i$  is the vector of covariates of the case in the  $i$ th stratum. In the general case of  $c_i$  cases per strata then the full conditional likelihood is

$$L = \prod_{i=1}^{n_s} \frac{\exp(s_i^T \beta)}{[\sum_{l \in C_i} \exp(s_l^T \beta)]} \quad (2)$$

where  $s_i$  is the sum of the vectors of covariates for the cases in the  $i$ th stratum and  $s_l$ ,  $l \in C_i$  refer to the sum of vectors of covariates for all distinct sets of  $c_i$  observations drawn from the  $i$ th stratum. The conditional likelihood can be maximized by a Newton-Raphson procedure. The covariances of the parameter estimates can be estimated from the inverse of the matrix of second derivatives of the logarithm of the conditional likelihood, while the first derivatives provide the score function,  $U_j(\beta)$ ,  $j = 1, 2, \dots, p$ , which can be used for testing the significance of parameters.

If the strata are not small,  $C_i$  can be large so to improve the speed of computation, the algorithm given by Howard in [3] and described by [4] is used.

A second situation in which the above conditional likelihood arises is in fitting Cox's proportional hazard model (see G12BAF) in which the strata refer to the risk sets for each failure time and the where failures are cases. When ties are present in the data G12BAF uses an approximation. For an exact estimate, the data can be expanded using G12ZAF to create the risk sets/strata and G11CAF used.

## 4 References

- [1] Cox D R (1972) Regression models in life tables (with discussion) *J. Roy. Statist. Soc. Ser. B* **34** 187–220
- [2] Cox D R and Hinkley D V (1974) *Theoretical Statistics* Chapman and Hall
- [3] Howard S (1972) Remark on the paper by Cox, D.R. (1972): Regression methods *J. R. Statist. Soc. B* **34** and life tables 187–220
- [4] Krailo M D and Pike M C (1984) Algorithm AS 196. Conditional multivariate *Appl. Statist.* **33** logistic analysis of stratified case-control studies 95–103
- [5] Smith P G, Pike M C, Hill P, Breslow N E and Day N E (1981) Multivariate conditional logistic analysis of stratum-matched case-control *Appl. Statist.* (Algorithm AS 162.) **30** studies 190–197

## 5 Parameters

- 1: N — INTEGER *Input*  
*On entry:* the number of observations,  $n$ .  
*Constraint:*  $N \geq 2$ .
- 2: M — INTEGER *Input*  
*On entry:* the number of covariates in array Z.  
*Constraint:*  $M \geq 1$ .
- 3: NS — INTEGER *Input*  
*On entry:* the number of strata,  $n_s$ .  
*Constraint:*  $NS \geq 1$ .
- 4: Z(LDZ,M) — *real* array *Input*  
*On entry:* the  $i$ th row must contain the covariates which are associated with the  $i$ th observation.
- 5: LDZ — INTEGER *Input*  
*On entry:* the first dimension of the array Z as declared in the (sub)program from which G11CAF is called.  
*Constraint:*  $LDZ \geq N$ .
- 6: ISZ(M) — INTEGER array *Input*  
*On entry:* indicates which subset of covariates are to be included in the model.  
 If  $ISZ(j) \geq 1$  the  $j$ th covariate is included in the model.  
 If  $ISZ(j) = 0$  the  $j$ th covariate is excluded from the model and not referenced.  
*Constraints:*  $ISZ(j) \geq 0$  and at least one value must be non-zero.

- 7:** IP — INTEGER *Input*  
*On entry:* the number of covariates included in the model,  $p$  as indicated by ISZ.  
*Constraints:*  $IP \geq 1$  and  $IP =$  number of non-zero values of ISZ.
- 8:** IC(N) — INTEGER array *Input*  
*On entry:* indicates whether the  $i$ th observation is a case or a control.  
 $IC(i) = 0$  indicates that the  $i$ th observation is a case.  
 $IC(i) = 1$  indicates that the  $i$ th observation is a control.  
*Constraint:*  $IC(i) = 0$  or  $1$  for  $i = 1, 2, \dots, N$ .
- 9:** ISI(N) — INTEGER array *Input*  
*On entry:* stratum indicators which also allow data points to be excluded from the analysis.  
 $ISI(i) = k$  indicates that the  $i$ th observation is from the  $k$ th stratum, where  $k = 1, 2, \dots, NS$ .  
 $ISI(i) = 0$  indicates that the  $i$ th observation is to be omitted from the analysis.  
*Constraints:*  $0 \leq ISI(i) \leq NS$  for  $i = 1, 2, \dots, N$ , and more than IP values of  $ISI(i) > 0$ .
- 10:** DEV — *real* *Output*  
*On exit:* the deviance, that is  $-2 \times (\text{maximized log marginal likelihood})$ .
- 11:** B(IP) — *real* array *Input/Output*  
*On entry:* initial estimates of the covariate coefficient parameters  $\beta$ .  $B(j)$  must contain the initial estimate of the coefficient of the covariate in  $Z$  corresponding to the  $j$ th non-zero value of ISZ.  
*Suggested values:* in many cases an initial value of zero for  $B(j)$  may be used. For another suggestion see Section 8.  
*On exit:*  $B(j)$  contains the estimate  $\hat{\beta}_i$  the coefficient of the covariate stored in the  $i$ th column of  $Z$  where  $i$  is the  $j$ th non-zero value in the array ISZ.
- 12:** SE(IP) — *real* array *Output*  
*On exit:*  $SE(j)$  is the asymptotic standard error of the estimate contained in  $B(j)$  and score function in  $SC(j)$  for  $j = 1, 2, \dots, IP$ .
- 13:** SC(IP) — *real* array *Output*  
*On exit:*  $SC(j)$  is the value of the score function  $U_j(\beta)$  for the estimate contained in  $B(j)$ .
- 14:** COV(IP\*(IP+1)/2) — *real* array *Output*  
*On exit:* the variance-covariance matrix of the parameter estimates in  $B$  stored in packed form by column, i.e., the covariance between the parameter estimates given in  $B(i)$  and  $B(j)$ ,  $j \geq i$ , is given in  $COV(j(j-1)/2 + i)$ .
- 15:** NCA(NS) — INTEGER array *Output*  
*On exit:*  $NCA(i)$  contains the number of cases in the  $i$ th stratum for  $i = 1, 2, \dots, NS$ .
- 16:** NCT(NS) — INTEGER array *Output*  
*On exit:*  $NCT(i)$  contains the number of controls in the  $i$ th stratum for  $i = 1, 2, \dots, NS$ .
- 17:** TOL — *real* *Input*  
*On entry:* indicates the accuracy required for the estimation. Convergence is assumed when the decrease in deviance is less than  $TOL \times (1.0 + \text{CurrentDeviance})$ . This corresponds approximately to an absolute accuracy if the deviance is small and a relative accuracy if the deviance is large.  
*Constraint:*  $TOL \geq 10 \times \text{machine precision}$ .

**18:** MAXIT — INTEGER *Input*

*On entry:* the maximum number of iterations required for computing the estimates. If MAXIT is set to 0 then the standard errors, the score functions and the variance-covariance matrix are computed for the input value of  $\beta$  in B but  $\beta$  is not updated.

*Constraint:* MAXIT  $\geq$  0.

**19:** IPRINT — INTEGER *Input*

*On entry:* indicates if the printing of information on the iterations is required. If IPRINT  $\leq$  0, there is no printing. If IPRINT  $\geq$  1, the deviance and the current estimates are printed every IPRINT iterations.

When printing occurs the output is directed to the current advisory message unit (see X04ABF).

**20:** WK(LWK) — *real* array *Workspace*

**21:** LWK — INTEGER *Input*

*On entry:* the length of the workspace array WK.

*Constraint:* LWK  $\geq pn_0 + (c_m + 1)(p + 1)(p + 2)/2 + c_m$ , where  $n_0$  is the number of observations included in the model, i.e., the number of observations for which  $ISI(i) \neq 0$  and  $c_m$  is the maximum number of observations in any stratum.

**22:** IFAIL — INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0,  $-1$  or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

## 6 Errors and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

- On entry, M < 1,
- or N < 2,
- or NS < 1,
- or IP < 1,
- or LDZ < N,
- or TOL <  $10 \times$  *machine precision*,
- or MAXIT < 0.

IFAIL = 2

- On entry, ISZ( $i$ ) < 0 for some  $i$ ,
- or the value of IP is incompatible with ISZ,
- or IC( $i$ )  $\neq$  1 or 0.
- or ISI( $i$ ) < 0 or ISI( $i$ ) > NS,
- or number of values of ISZ( $i$ ) > 0 is greater than or equal to  $n_0$ , the number of observations excluding any with ISI( $i$ ) = 0.

IFAIL = 3

The value of LWK is too small.

IFAIL = 4

Overflow has been detected. Try using different starting values.

IFAIL = 5

The matrix of second partial derivatives is singular. Try different starting values or include fewer covariates.

IFAIL = 6

Convergence has not been achieved in MAXIT iterations. The progress towards convergence can be examined by using a non-zero value of IPRINT. Any non-convergence may be due to a linear combination of covariates being monotonic with time.

Full results are returned.

## 7 Accuracy

The accuracy is specified by TOL.

## 8 Further Comments

The other models described in Section 3 can be fitted using the generalized linear modelling routines G02GBF and G02GCF.

The case with one case per stratum can be analysed by having a dummy response variable  $y$  such that  $y = 1$  for a case and  $y = 0$  for a control, and fitting a Poisson generalized linear model with a log link and including a factor with a level for each strata. These models can be fitted by using G02GCF.

G11CAF uses mean centering which involves subtracting the means from the covariables prior to computation of any statistics. This helps to minimize the effect of outlying observations and accelerates convergence. In order to reduce the risk of the sums computed by Howard's algorithm becoming too large, the scaling factor described in [4] is used.

If the initial estimates are poor then there may be a problem with overflow in calculating  $\exp(\beta^T z_i)$  or there may be non-convergence. Reasonable estimates can often be obtained by fitting an unconditional model.

## 9 Example

The data was used for illustrative purposes by [5] and consists of two strata and two covariates. The data is input, the model is fitted and the results are printed.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G11CAF Example Program Text.
*      Mark 19 Release. NAG Copyright 1999.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, MMAX, SMAX, MLWK
      PARAMETER        (NMAX=500,MMAX=20,SMAX=10,MLWK=10000)
*      .. Local Scalars ..
      real            DEV, TOL
      INTEGER          I, IFAIL, IP, IPRINT, J, LDZ, LWK, M, MAXIT, N,
+                   NS
*      .. Local Arrays ..
```

```

      real          B(MMAX), COV(MMAX*(MMAX+1)/2), SC(MMAX),
+                SE(MMAX), WK(MLWK), Z(NMAX,MMAX)
      INTEGER      IC(NMAX), ISI(NMAX), ISZ(MMAX), NCA(SMAX),
+                NCT(SMAX)
*      .. External Subroutines ..
      EXTERNAL     G11CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G11CAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      LDZ = NMAX
      LWK = MLWK
      TOL = 1.0e-5
      READ (NIN,*) N, M, NS, MAXIT, IPRINT
      IF (N.LE.NMAX .AND. M.LE.MMAX .AND. NS.LE.SMAX) THEN
        DO 20 I = 1, N
          READ (NIN,*) ISI(I), IC(I), (Z(I,J),J=1,M)
20      CONTINUE
          READ (NIN,*) (ISZ(J),J=1,M), IP
          READ (NIN,*) (B(J),J=1,M)
*
          IFAIL = 0
*
          CALL G11CAF(N,M,NS,Z,LDZ,ISZ,IP,IC,ISI,DEV,B,SE,SC,COV,NCA,NCT,
+                TOL,MAXIT,IPRINT,WK,LWK,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,99999) ' Deviance = ', DEV
          WRITE (NOUT,*)
          WRITE (NOUT,*) ' Strata      No. Cases   No. Controls'
          WRITE (NOUT,*)
          DO 40 I = 1, NS
            WRITE (NOUT,99998) I, NCA(I), NCT(I)
40      CONTINUE
          WRITE (NOUT,*)
          WRITE (NOUT,*) ' Parameter      Estimate',
+                ' Standard Error'
          WRITE (NOUT,*)
          DO 60 I = 1, IP
            WRITE (NOUT,99997) I, B(I), SE(I)
60      CONTINUE
          END IF
*
99999 FORMAT (A,e13.4)
99998 FORMAT (3X,I3,2(10X,I2))
99997 FORMAT (I6,2(10X,F8.4))
      END

```

## 9.2 Program Data

G11CAF Example Program Data

7 2 2 10 0

1 0 0 1

1 0 1 2

1 1 0 1

```

1 1 1 3
2 0 0 1
2 1 1 0
2 1 0 2

    1 1 2

0.0 0.0

```

### 9.3 Program Results

G11CAF Example Program Results

Deviance = 0.5475E+01

Strata	No. Cases	No. Controls
1	2	2
2	1	2

Parameter	Estimate	Standard Error
1	-0.5223	1.3901
2	-0.2674	0.8473

---